KIT
de sobrevivencia em
CALCULO
Prof. Doherty Andrade - DMA- UEM
doherty@gauss.dma.uem.br

Integrais Múltiplas em várias coordenadas

Mudanca de variaveis, coordenadas cilindricas e esfericas .

Seja W uma região no espaço tridimensional e f definida em W. Se W pode ser definido em termos de coordenadas retangulares, então a integral sobre a região pode avaliada por meio de uma integral tripa iterada.

Como exemplo vamos plotar a regiao W que é a interseção de dois cilindros circulares retos de raio 1. Um é simetrico em relação eixo x e o outro é simetrico em relação ao eixo z. Os seguintes comandos plotam os cilindros.

```
> restart;

> W1 := plot3d([x,cos(theta),sin(theta)],theta=0..2*Pi,x=-1..1,

> scaling =constrained):

> W2 := plot3d([cos(theta),sin(theta),z],theta=0..2*Pi,z=-1..1,

> scaling =constrained):

> with(plots):

> display3d({W1,W2});
```



>

Vamos determinar o volume do solido W formado pela intersecao destes dois cilindros. Notemos que W eh composto de oito regioes identicas, assim vamos nos concentrar na parte que estah no primeiro octante. Vamos olhar esta regiao.

```
> W3 := plot3d([x,cos(theta),sin(theta)],theta=0..Pi/2,x=0..1,
> scaling =constrained,style=wireframe):
> W4 := plot3d([cos(theta),sin(theta),z],theta=0..Pi/2,z=0..1,
> scaling =constrained,style=wireframe):
> display3d({W3,W4});
```



>

O volume desta parte eh

```
> ParteVolume:=Int(Int(Int(1,z=0..sqrt(1-y^2)),x=0..sqrt(1-y^2)),
> y=0..1);
```

$$ParteVolume := \int_0^1 \int_0^{\sqrt{1-y^2}} \int_0^{\sqrt{1-y^2}} 1 \, dz \, dx \, dy$$

```
> ParteVolume := value(ParteVolume);
```

$$ParteVolume := \frac{2}{3}$$

e assim o volume do solido inteiro eh oito vezes o volume da parte ja calculada.

```
> TotalVolume := 8*ParteVolume;
```

$$TotalVolume := \frac{16}{3}$$

Usamos integral tripla para calcular o volume, mas poderiamos ter usado integral dupla.

```
> ParteVolume := Int(Int(sqrt(1-y^2), x=0..sqrt(1-y^2)),y=0..1);
```

$$ParteVolume := \int_0^1 \int_0^{\sqrt{1-y^2}} \sqrt{1-y^2}\, dx\, dy$$

> **ParteVolume := value(ParteVolume);**

$$ParteVolume := \frac{2}{3}$$

Agora vamos resolver outro problema com o mesmo solido. Seja T a superficie superior do solido no primeiro octante e suponha que a densidade de T em qualquer ponto (x,y,z) eh igual ao quadrado da distancia do ponto ao plano z = 0, i.e.,

> **rho(x,y,z) = z^2;**

$$\rho(x, y, z) = z^2$$

> **Tmass := 4*Int(Int(Int(z^2,z=0..sqrt(1-y^2)),x=0..sqrt(1-y^2)),**

> **y=0..1);**

$$Tmass := 4 \int_0^1 \int_0^{\sqrt{1-y^2}} \int_0^{\sqrt{1-y^2}} z^2\, dz\, dx\, dy$$

> **Tmass := value(Tmass);**

$$Tmass := \frac{32}{45}$$

Agora vamos encontrar o centro de massa de T , devemos encontrar o momento de T com relacao ao plano z=0.

> **Tmomentoxy := 4*Int(Int(Int(z^3,z=0..sqrt(1-y^2)),x=0..sqrt(1-y^2)),**

> **y=0..1);**

$$Tmomentoxy := 4 \int_0^1 \int_0^{\sqrt{1-y^2}} \int_0^{\sqrt{1-y^2}} z^3\, dz\, dx\, dy$$

> **Tmomentoxy := value(Tmomentoxy);**

$$Tmomentoxy := \frac{5}{32}\pi$$

> **zbar := Tmomentoxy/Tmass;**

$$zbar := \frac{225}{1024}\pi$$

Assim o centro de massa de T estah localizado em ($0, 0, \frac{225\pi}{1024}$).

Muitas vezes eh conveniente usar coordenadas polares quando calculamos integrais duplas sobre regioes apropriadas, tambem eh util usar coordenadas cilindricas no calculo de integrais triplas.

A relacao entre coordendas cilindricas e coordenadas retangulares eh

$x = r\cos(\theta)$, $y = r\sin(\theta)$, $z = z$.

Quando mudamos as coordenadas devemos determinar o que usar no lugar de dV. Isto eh feito usando o Jacobiano.

> **with(linalg):**

Warning, new definition for norm

Warning, new definition for trace

> **JacobianoC := jacobian([r*cos(theta),r*sin(theta),z],[r,theta,z]);**

$$JacobianoC := \begin{bmatrix} \cos(\theta) & -r\sin(\theta) & 0 \\ \sin(\theta) & r\cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

> **DetjacobianoC := simplify(det(JacobianoC));**

$$DetjacobianoC := r$$

Isto significa que quando mudamos as coordenas para coordenadas cilindricas usamos dV dado por

$dV = r\,dz\,dr\,d\theta$.

>

Como exemplo cosidere o solido W abaixo do plane $z = 1$ e o semi cone superior circular reto

$$z = \sqrt{x^2 + y^2}$$.

Vamos plotar este cone usando o comando **cylinderplot** .

```
> with(plots):
```

```
> cylinderplot(z,theta = 0..2*Pi,z=0..1);
```



\>

O volume de W usando coordenadas cilindricas eh

```
> Volume := Int(Int(Int(r,z=r..1),r=0..1),theta=0..2*Pi);
```

$$Volume := \int_0^{2\pi} \int_0^1 \int_r^1 r \, dz \, dr \, d\theta$$

```
> Volume := value(Volume);
```

$$Volume := \frac{1}{3}\pi$$

Como outro exemplo, vamos converter a seguinte integral tripla dada em coordenadas retangulares para coordenadas cilindricas :

```
> Int(Int(Int(x^2+y^2,z=0..4-x^2-y^2),y=-sqrt(4-x^2)..sqrt(4-x^2)),
```

> **z=-2..2);**

$$\int_{-2}^{2}\int_{-\sqrt{4-x^2}}^{\sqrt{4-x^2}}\int_{0}^{4-x^2-y^2} x^2+y^2\,dz\,dy\,dz$$

Esta integral tripla da funcao $f(x,y,z)=x^2+y^2$ sobre o solido W limitado acima pelo pelo paraboloide $z=4-x^2-y^2$, $0\le z$, e limitado abaixo pelo plano $z=0$. Vamos plotar esta regiao com o comando **cylinderplot** .

> **cylinderplot(sqrt(4-z),theta=0..2*Pi,z=0..4,scaling=constrained);**



>

A integral em coordenadas cilindricas fica

> **V := Int(Int(Int(r^2*r,z=0..4-r^2),r=0..2),theta=0..2*Pi);**

$$V:=\int_{0}^{2\pi}\int_{0}^{2}\int_{0}^{4-r^2} r^3\,dz\,dr\,d\theta$$

> **V:= value(V);**

$$V := \frac{32}{3}\pi$$

Coordenadas esfericas estao relacionadas com as coordenadas retangulares como segue:

$x = \rho \sin(\phi)\cos(\theta)$, $y = \rho \sin(\phi)\sin(\theta)$, $z = \rho \cos(\phi)$.

Usando o determinante do Jacobiano desta transformacao obtemos a expresso para dV.

> **JacobianoS :=jacobian(**

> **[rho*sin(phi)*cos(theta),rho*sin(phi)*sin(theta),**

> **rho*cos(phi)],[rho,phi,theta]);**

$$JacobianoS := \begin{bmatrix} \sin(\phi)\cos(\theta) & \rho\cos(\phi)\cos(\theta) & -\rho\sin(\phi)\sin(\theta) \\ \sin(\phi)\sin(\theta) & \rho\cos(\phi)\sin(\theta) & \rho\sin(\phi)\cos(\theta) \\ \cos(\phi) & -\rho\sin(\phi) & 0 \end{bmatrix}$$

> **DetJacobianoS := simplify(det(JacobianoS));**

$$DetJacobianoS := \sin(\phi)\,\rho^2$$

Assim em coordenadas esfericas temos

$dV = \rho^2 \sin(\phi)\, d\rho\, d\phi\, d\theta$.

Como exemplo considere o solido limitado abaixo pelo semi cone superior $\phi = \frac{\pi}{3}$

e acima pela esfera $\rho = 1$.

Primeiro plotamos esta regiao usando **sphereplot** e **cylinderplot** .

> **P1 := sphereplot(1,theta = 0..2*Pi,phi=0..Pi/2):**

Como antes obtemos o cone como segue:

> **P2 := cylinderplot(z,theta=0..2*Pi,z=0..1):**

O comando que plota as duas superficies ao mesmo tempo eh :

> **display3d({P1,P2},style =wireframe,axes=normal);**



>

O volume deste solido eh dado pela seguinte integral tripla:

> **V := Int(Int(Int(rho^2*sin(phi),rho=0..1),phi=0..Pi/3),**

> **theta=0..2*Pi);**

$$V := \int_0^{2\pi} \int_0^{\frac{1}{3}\pi} \int_0^1 \sin(\phi)\,\rho^2 \,d\rho\,d\phi\,d\theta$$

> **V := value(V);**

$$V := \frac{1}{3}\pi$$

>

Exercicios

1. Determine o volume do solido tridimensional limitado pelas supeficies

$z = x^2 + 4y^2$, e $z = 12 - 2x^2 - 2y^2$.

2. Determine a massa do solido limitado pelos planos y=0 e $z = 0$ e as superficies $z = 4 - x^2$ e $x = y^2$ assumindo que a funcao densidade eh dada por $\delta(x, y, z) = x y^2 z^3$.